

# dati.senato.it

## Open Data del Senato della Repubblica

Carlo Marchetti e Giovanni Lalle  
Senato della Repubblica  
Servizio dell'Informatica

# Servizi disponibili

- Dati: SPARQL endpoint + sito web
- Documenti:
  - DDL e Dossier (Akoma Ntoso XML)
  - Risolutore di URN

# Dati

- [dati.senato.it/sparql](http://dati.senato.it/sparql)
  - 34M triple RDF
  - 24 classi OSR (riuso di OCD, FOAF, BIO)
  - aggiornati dalla XIII Leg. (1996) a ieri
  - licenza CCBY

**Demo**

# Documenti

- Akoma Ntoso: standard per il markup dei documenti legislativi
  - [www.akomantoso.org](http://www.akomantoso.org)
  - Schema Senato disponibile all'indirizzo [www.senato.it/download/public/akomantoso30.xsd](http://www.senato.it/download/public/akomantoso30.xsd)
- OASIS TC
- Evoluzione: migliore legame tra dati e documenti

<http://dati.senato.it/sparql>

osr:Senatore

osr:Commissione

ocd:GruppoParlamentare

osr:SedutaCommissione

osr:ConsiglioDiPresidenza

osr:SedutaAssemblea

osr:Ddl

osr:Documento

osr:SindacatoIspettivo


osr:Intervento

osr:Relatore

osr:Votazione

osr:Iniziativa

# Senatori




# Senatori

## Tutti i senatori della XVII Leg.

```
SELECT DISTINCT ?senatore ?foto ?nome ?cognome ?legislatura
?inizioMandato ?fineMandato ?tipoMandato
?tipoFineMandato ?dataNascita ?cittaNascita
?provinciaNascita ?nazioneNascita
WHERE {
    ?senatore a osr:Senatore.
    ?senatore foaf:firstName ?nome.
    ?senatore foaf:lastName ?cognome.
    ?senatore foaf:depiction ?foto.
    ?senatore osr:cittaNascita ?cittaNascita.
    ?senatore osr:provinciaNascita ?provinciaNascita.
    ?senatore osr:nazioneNascita ?nazioneNascita.
    ?senatore osr:dataNascita ?dataNascita.
    ?senatore osr:mandato ?mandato.
    ?mandato osr:tipoMandato ?tipoMandato.
    ?mandato osr:inizio ?inizioMandato.
    OPTIONAL { ?mandato osr:fine ?fineMandato. }
    OPTIONAL { ?mandato osr:tipoFineMandato ?tipoFineMandato. }
    ?mandato osr:legislatura ?legislatura.
    FILTER(?legislatura=17)
} ORDER BY ?cognome ?nome
```



# Gruppi




# Gruppi

## Composizione attuale dei Gruppi

```
SELECT DISTINCT ?gruppo ?nomeGruppo ?senatore
?nome ?cognome ?carica ?inizioAdesione
WHERE
{
  ?gruppo a ocd:gruppoParlamentare.
  ?gruppo osr:denominazione ?denominazione.
  ?denominazione osr:titolo ?nomeGruppo.
  ?adesioneGruppo a ocd:adesioneGruppo.
  ?adesioneGruppo osr:carica ?carica.
  ?adesioneGruppo osr:inizio ?inizioAdesione.
  ?adesioneGruppo osr:gruppo ?gruppo.
  ?senatore ocd:aderisce ?adesioneGruppo.
  ?senatore a osr:Senatore.
  ?senatore foaf:firstName ?nome.
  ?senatore foaf:lastName ?cognome.
  OPTIONAL { ?adesioneGruppo osr:fine ?fineAdesione }
  OPTIONAL { ?denominazione osr:fine ?fineDenominazione }
  FILTER(!bound(?fineAdesione) && !bound(?fineDenominazione))
}
GROUP BY ?nomeGruppo
ORDER BY ?nomeGruppo
```

# Commissioni




# Commissioni

## Composizione attuale delle Commissioni

```
SELECT DISTINCT ?senatore ?nome ?cognome
?inizioAfferenza ?carica ?commissione ?dizione
WHERE
{
    ?senatore osr:afferisce ?afferenza.
    ?senatore foaf:firstName ?nome.
    ?senatore foaf:lastName ?cognome.
    ?senatore a osr:Senatore.
    ?afferenza osr:inizio ?inizioAfferenza.
    ?afferenza osr:commissione ?commissione.
    ?afferenza osr:carica ?carica.
    ?commissione osr:denominazione ?denominazione.
    ?denominazione osr:titolo ?dizione.
    OPTIONAL {?denominazione osr:fine ?dataFine}
    OPTIONAL {?afferenza osr:fine ?fineAfferenza}
    FILTER(!bound(?fineAfferenza))
    FILTER(!bound(?dataFine))
}
GROUP BY ?commissione
ORDER BY ?commissione ?cognome ?nome
```

# Disegni di legge (DDL)



# Disegni di legge (DDL)

DDL presentati dal 1/1/2014 al 30/4/2014

```
SELECT DISTINCT ?ddl ?idFase ?ramo ?legislatura ?numeroFase ?titolo
?titoloBreve ?natura ?stato ?dataStato ?dataPresentazione
?presentatoTrasmesso ?testoPresentato ?testoApprovato
WHERE {
  ?ddl a osr:Ddl.
  ?ddl osr:idFase ?idFase.
  ?ddl osr:statoDdl ?stato.
  ?ddl osr:ramo ?ramo.
  ?ddl osr:dataPresentazione ?dataPresentazione.
  ?ddl osr:titolo ?titolo.
  OPTIONAL { ?ddl osr:titoloBreve ?titoloBreve }.
  ?ddl osr:presentatoTrasmesso ?presentatoTrasmesso.
  ?ddl osr:natura ?natura.
  ?ddl osr:dataStatoDdl ?dataStato.
  ?ddl osr:numeroFase ?numeroFase.
  ?ddl osr:numeroFaseCompatto ?numeroFaseCompatto.
  ?ddl osr:legislatura ?legislatura.
  ?ddl osr:legislatura 17 .
  OPTIONAL { ?ddl osr:testoPresentato ?testoPresentato }
  OPTIONAL { ?ddl osr:testoApprovato ?testoApprovato }
  FILTER(xsd:date(str(?dataPresentazione)) <= xsd:date(str("2014-04-30")))
  FILTER(xsd:date(str(?dataPresentazione)) >= xsd:date(str("2014-01-01")))
}
ORDER BY ?ramo ?numeroFaseCompatto
```

# URN

## Uniform Resource Name

### Risolutore

`http://www.senato.it/uri-res/N2Ls?<URN>`

### Testo presentato

`urn:nir:senato.repubblica:diseagno.legge:<LEG>.legislatura;<NUMERO>`

`http://www.senato.it/uri-res/N2Ls?urn:nir:senato.repubblica:diseagno.legge:17.legislatura;1470`

### Testo approvato

`urn:nir:senato.repubblica:diseagno.legge;approvato:<LEG>.legislatura;<NUMERO>`

`http://www.senato.it/uri-res/N2Ls?urn:nir:senato.repubblica:diseagno.legge;approvato:17.legislatura;1470`

# Disegni di legge (DDL)

## Iniziativa del DDL S.687 (cittadinanza)

```
SELECT DISTINCT ?presentatore ?flagPrimoFirmatario
?dataAggiuntaFirma ?dataRitiroFirma
WHERE {
  ?ddl a osr:Ddl.
  ?ddl osr:idFase ?idFase.
  ?ddl osr:statoDdl ?stato.
  ?ddl osr:ramo ?ramo.
  ?ddl osr:dataPresentazione ?dataPresentazione.
  ?ddl osr:titolo ?titolo.
  OPTIONAL { ?ddl osr:titoloBreve ?titoloBreve }.
  ?ddl osr:presentatoTrasmesso ?presentatoTrasmesso.
  ?ddl osr:natura ?natura.
  ?ddl osr:dataStatoDdl ?dataStato.
  ?ddl osr:numeroFase ?numeroFase.
  ?ddl osr:numeroFaseCompatto ?numeroFaseCompatto.
  OPTIONAL { ?ddl osr:testoPresentato ?testoPresentato }
  OPTIONAL { ?ddl osr:testoApprovato ?testoApprovato }
  ?ddl osr:iniziativa ?iniziativa .
  ?iniziativa osr:presentatore ?presentatore .
  OPTIONAL { ?iniziativa osr:primoFirmatario ?flagPrimoFirmatario }
  OPTIONAL { ?iniziativa osr:dataRitiroFirma ?dataRitiroFirma }
  OPTIONAL { ?iniziativa osr:dataAggiuntaFirma ?dataAggiuntaFirma }
  ?ddl osr:idFase 41007 .
}
ORDER BY DESC(?flagPrimoFirmatario)
```




# Disegni di legge (DDL)

Iter del DDL S.1464

(DL 34/2014 - rilancio occupazione)

```
SELECT DISTINCT ?progrIter ?fase ?ramo ?numeroFase
?stato ?dataStato
WHERE {
    ?ddl a osr:Ddl.
    ?ddl osr:idDdl ?idDdl.
    ?iter a osr:IterDdl.
    ?iter osr:idDdl ?idDdl.
    ?iter osr:fase ?faseIter.
    ?faseIter osr:ddl ?fase.
    ?faseIter osr:progrIter ?progrIter.
    ?fase osr:ramo ?ramo.
    ?fase osr:numeroFase ?numeroFase.
    ?fase osr:statoDdl ?stato.
    ?fase osr:dataStatoDdl ?dataStato.
    FILTER (?ddl = <http://dati.senato.it/ddl/44347> )
} ORDER BY ?progrIter
```

# Votazioni



# Votazioni

## Ultime 100 votazioni elettroniche

```
SELECT DISTINCT ?dataSeduta ?numeroSeduta ?votazione ?numero
?oggetto ?esito ?presenti ?maggioranza ?numeroLegale ?votanti
?favorevoli ?contrari ?astenuti ?presidente
WHERE {
    ?votazione a osr:Votazione.
    ?votazione osr:numero ?numero.
    ?votazione osr:seduta ?seduta.
    ?votazione osr:presenti ?presenti.
    ?votazione osr:favorevoli ?favorevoli.
    ?votazione osr:maggioranza ?maggioranza.
    ?votazione osr:votanti ?votanti.
    ?votazione osr:contrari ?contrari.
    ?votazione rdfs:label ?oggetto.
    ?seduta osr:numeroSeduta ?numeroSeduta.
    ?seduta osr:dataSeduta ?dataSeduta.
    ?seduta osr:legislatura 17 .
    ?votazione osr:esito ?esito.
    ?votazione osr:astenuti ?astenuti.
    ?votazione osr:numeroLegale ?numeroLegale.
    ?votazione osr:presidente ?presidente
}
ORDER BY DESC(?dataSeduta) DESC(?numeroSeduta) DESC(?numero)
LIMIT 100
```

Questa presentazione è disponibile all'indirizzo

<http://www.senato.it/download/public/hackathon.pdf>

**Grazie**